query: embed.html#
  SELECT
    (SAMPLE(?number_of_works_) AS ?works)
    ?researcher ?researcherLabel ?researcherDescription
    (SAMPLE(?orcid_) AS ?orcid) 
    (SAMPLE(?scopus_) AS ?scopus)
    (SAMPLE(?researcherid_) AS ?researcherid)
  WITH {
    SELECT DISTINCT ?researcher WHERE {
      ?researcher ( wdt:P108 | wdt:P463 | wdt:P1416 | wdt:P69 ) / wdt:P361* wd:Q503424 . 
    } 
  } AS %researchers
  WITH {
    SELECT
      (COUNT(?work) AS ?number_of_works_) ?researcher
    WHERE {
      INCLUDE %researchers
      OPTIONAL { ?work wdt:P50 ?researcher . }
    } 
    GROUP BY ?researcher
  } AS %researchers_and_number_of_works
  WHERE {
    INCLUDE %researchers_and_number_of_works
    OPTIONAL { ?researcher wdt:P496 ?orcid_ . }
    OPTIONAL { ?researcher wdt:P1153 ?scopus_ . }
    OPTIONAL { ?researcher wdt:P1053 ?researcherid_ . }
    SERVICE wikibase:label { bd:serviceParam wikibase:language "en,da,de,es,fr,nl,no,ru,sv,zh" . } 
  }
  GROUP BY ?researcher ?researcherLabel ?researcherDescription 
  ORDER BY DESC(?works)
   